

21世纪高等学校计算机规划教材

21st Century University Planned Textbooks of Computer Science

Visual Basic 程序设计实践教程

Experiment Instructions for Visual Basic
Programming Language

李雁翎 夏龙 涂美彩 编著

- 构建程序设计知识体系
- 强调程序设计思想方法技巧
- 培养应用程序开发理念



名家系列

 人民邮电出版社
POSTS & TELECOM PRESS

21世纪高等学校计算机规划教材

21st Century University Planned Textbooks of Computer Science

Visual Basic 程序设计实践教程

Experiment Instructions for Visual Basic
Programming Language

李雁翎 夏龙 涂美彩 编著



名家系列

人民邮电出版社

北京

图书在版编目 (CIP) 数据

Visual Basic 程序设计实践教程 / 李雁翎, 夏龙, 涂美彩编
著. —北京: 人民邮电出版社, 2008.5
21 世纪高等学校计算机规划教材
ISBN 978-7-115-17655-4

I. V… II. ①李…②夏…③涂… III. BASIC 语言—程
序设计—高等学校—教学参考资料 IV. TP312

中国版本图书馆 CIP 数据核字 (2008) 第 020987 号

内 容 提 要

本书是《Visual Basic 程序设计教程》(普通高等教育“十一五”国家级规划教材)一书配套的辅助教材。全书共分 3 篇: 第 1 篇为实验指导篇, 根据主教材 4~15 章介绍的相关内容, 编排了 11 个综合实验题目, 详细介绍每一个实验的实验目的、实验手段及实验方法; 第 2 篇为综合实例篇, 拓展了编程的思维方式, 体现了 Visual Basic 应用程序的功能的强大; 第 3 篇为习题解答篇, 是按照主教材各章的习题而编写的习题解答, 一题一解, 便于对主教材相关知识点的理解和检验。

本书习题内容解答详细, 力求针对性强; 实验内容丰富, 综合性强; 综合实例对各章节的知识点加以适当扩充, 其应用性相对于主教材例题有所提升, 有利于学生知识的掌握和实践能力的提高。

本书可作为高等院校非计算机专业“Visual Basic 程序设计”课程配套的实验指导教材, 也可供大中专院校计算机专业使用。

21 世纪高等学校计算机规划教材

Visual Basic 程序设计实践教程

- ◆ 编 著 李雁翎 夏 龙 涂美彩
责任编辑 张 鑫
- ◆ 人民邮电出版社出版发行 北京市崇文区夕照寺街 14 号
邮编 100061 电子函件 315@ptpress.com.cn
网址 <http://www.ptpress.com.cn>
北京顺义振华印刷厂印刷
新华书店总店北京发行所经销
- ◆ 开本: 787×1092 1/16
印张: 11.75
字数: 309 千字 2008 年 5 月第 1 版
印数: 1—3 000 册 2008 年 5 月北京第 1 次印刷

ISBN 978-7-115-17655-4/TP

定价: 19.80 元

读者服务热线: (010)67170985 印装质量热线: (010)67129223
反盗版热线: (010)67171154

前言

《Visual Basic 程序设计教程》(普通高等教育“十一五”国家级规划教材)一书自 2007 年 2 月出版以来,受到了广大读者的欢迎。应读者的要求及建议,作者总结近年的教学实践,并结合从事 Visual Basic 教学的切身体会,编写了这本《Visual Basic 程序设计实践教程》,供广大读者在学习或开发实践中参考使用。

全书共分为 3 篇,第 1 篇是实验指导,第 2 篇是综合实例,第 3 篇是习题解答。

实验指导是根据主教材第 4~15 章介绍的相关内容,编排了 11 个综合实验题目,详细介绍了每一个实验的实验目的、实验手段及实验方法。11 个综合实验题目是根据主教材的内容精心编排、设计的。通过每个综合实验对主教材相关章节的内容加以消化和理解,并对各章节的知识点做了适当的扩充,使实验的应用性、综合性相对主教材例题有所提升,有利于对主教材知识点的掌握和实践能力的提高。11 个综合实验题目分别是:指针式时钟、参赛队对阵表生成器、打印字符图形、交换变量的值、复制文件、选课系统、网页浏览器、射击特训、超市销售管理、MP3 播放器、透明窗体。综合实验都是按照实验目的、实验要求,从窗体的组成、窗体中每个控件的属性,到每个控件的事件代码的顺序,逐项加以介绍,内容完整,易学易操作。

综合实例是根据 Visual Basic 程序的特点而编写的两个大型实例,分别是日志管理和简单绘图程序。综合实例中尽量应用了教材中涉及的所有知识点,有利于提高学生的综合编程能力。

习题解答是针对《Visual Basic 程序设计教程》一书各章的习题而作的解答,一题一解。通过解答对主教材中概念、知识点做详细温习,并注意对程序设计类习题解题的方法和步骤做出详细的讲解,从培养学生创造性思维入手,加强程序设计方法及算法分析的内容比重,增强学生分析问题、解决问题的能力。

本书由李雁翎、夏龙、涂美彩编写,由李雁翎统筹设计并统稿,东北师范大学软件工程专业 06 级的部分同学参与了对本书内容的讨论,并提供了良好的建议,在此表示衷心感谢。

由于编写时间有限,书中难免有错误和不足之处,敬请广大读者批评指正。

编者

2008 年 1 月

目 录

第 1 篇 实验指导	1
实验 1 指针式时钟	1
实验 2 参赛队对阵表生成器	4
实验 3 打印字符图形	6
实验 4 交换变量的值	8
实验 5 复制文件	9
实验 6 选课系统	16
实验 7 网页浏览器	19
实验 8 射击特训	31
实验 9 超市销售管理	34
实验 10 MP3 播放器	36
实验 11 透明窗体	40
第 2 篇 综合实例	43
实例 1 日志管理	43
实例 2 简单绘图程序	64
第 3 篇 习题解答	82
习题 1 绪论	82
习题 2 程序设计基础	84
习题 3 面向对象程序设计基础	88
习题 4 窗体及基本的内部控件	91
习题 5 程序控制结构	97
习题 6 数组及应用	109
习题 7 过程及应用	123
习题 8 文件管理	133
习题 9 常用的内部控件	144
习题 10 ActiveX 控件	153
习题 11 绘图语句	159
习题 12 数据库控件	166
习题 13 多媒体控件	174
习题 14 API 函数	179
习题 15 应用程序设计	182

第 1 篇

实验指导

实验 1 指针式时钟

实验题目：设计一个“指针式时钟”程序。

该程序有如下控制功能：

- (1) 利用 shape 控件、line 控件及 timer 控件实现指针式电子时钟的功能；
- (2) 简单实现时间设置、背景色及前景色的设置。

程序的运行结果如图 1-1-1 所示。




图 1-1-1 指针式时钟

操作步骤如下。

- (1) “指针式时钟”窗体及主要控件属性参照表 1-1-1 设计。

表 1-1-1 “指针式时钟”窗体及主要控件的属性

对 象	对 象 名	属 性 名	属 性 值	事 件 名
窗体	Frm	Caption	指针式时钟	Load
		Height	4530	
		Width	5790	
形状	Shp	BackColor	&H80000005&	无
		Height	2775	
		Width	2895	
		BorderStyle	1 - Solid	
		Shape	3 - Circle	

续表

对 象	对 象 名	属 性 名	属 性 值	事 件 名
线	LinMM	BorderStyle	1 - Solid	无
		X1	2160	
		X2	2160	
		Y1	1320	
		Y2	2040	
	LinHH	BorderStyle	1 - Solid	无
		X1	2520	
		X2	2520	
		Y1	1440	
		Y2	2160	
	LinSS	BorderStyle	1 - Solid	无
		X1	2760	
		X2	2760	
		Y1	840	
		Y2	2160	
	Linret	BorderStyle	1 - Solid	无
		X1	3000	
		X2	3120	
		Y1	840	
		Y2	840	
时钟	Tmr	Enabled	True	Timer
		Interval	20	

(2) 打开“代码设计”窗口，输入程序代码。

定义窗体变量代码如下：

```
Option Explicit
Dim pi As Double
Dim reticle As Single
Dim alarmtime As String
Dim setvalue As String
```

'控制指针转动
'设置闹钟
'设置时间

Form_Load()事件代码如下：

```
Private Sub Form_Load()
    Dim x As Single, y As Single
    Dim i As Integer
    pi = 3.141592653
    Frm.Width = 3500
    Frm.Height = 4500
    Frm.Left = 5000
    Shp.Width = 2600
    Shp.Height = 2500
    Shp.Left = 400
    Shp.Top = 650
    Shp.BorderWidth = 2
    LinHH.BorderWidth = 3
    LinMM.BorderWidth = 2
    x = Frm.ScaleWidth / 2
    y = Frm.ScaleHeight / 2
    Scale (-x, y)-(x, -y)
```

'窗口初始化
'建立坐标系

```

For i = 1 To 12                                '每 30° 画一个刻度线
    reticle = pi * (i - 1) / 6
    Load Linret(i)
        Linret(i).X1 = 950 * Cos(reticle)
        Linret(i).Y1 = 950 * Sin(reticle)
        Linret(i).X2 = 1150 * Cos(reticle)
        Linret(i).Y2 = 1150 * Sin(reticle)
        Linret(i).Visible = True
        Linret(i).BorderWidth = 3
    Next i
    Tmr.Enabled = True
End Sub
Alarm_Click()事件代码如下:
Private Sub Alarm_Click()                      '闹钟设置
    alarmtime = InputBox("输入闹钟时间: ", "设置闹钟", Time)
End Sub
Settime_Click()事件代码如下:
Private Sub Settime_Click()                   '时间设置
    setvalue = InputBox("按标准格式输入时间", "设置时间", Time)
    If setvalue <> "" Then
        Time = setvalue
    End If
End Sub
Tmr_Timer()事件代码如下:
Private Sub Tmr_Timer()                      '控制指针转动
    Dim hh As Integer, mm As Integer, ss As Integer
    If Time = alarmtime Then                 '闹钟启动
        MsgBox "该起床了!", 64, "响铃"
    End If
    hh = Hour(Time)                          '当前系统时间值赋给变量
    mm = Minute(Time)
    ss = Second(Time)
    reticle = ss * 2 * pi / 60 + pi / 2      '控制秒针转动
    LinSS.X1 = 100 * Cos(reticle - pi)
    LinSS.Y1 = -100 * Sin(reticle - pi)
    LinSS.X2 = 1000 * Cos(reticle - pi)
    LinSS.Y2 = -1000 * Sin(reticle - pi)
    reticle = mm * 2 * pi / 60 + pi / 2      '控制分针转动
    LinMM.X1 = 100 * Cos(reticle)
    LinMM.Y1 = -100 * Sin(reticle)
    LinMM.X2 = 800 * Cos(reticle - pi)
    LinMM.Y2 = -800 * Sin(reticle - pi)
    reticle = hh * 2 * pi / 12 + mm * pi / 360 + pi / 2 '控制时针转动
    LinHH.X1 = 100 * Cos(reticle)
    LinHH.Y1 = -100 * Sin(reticle)
    LinHH.X2 = 600 * Cos(reticle - pi)
    LinHH.Y2 = -600 * Sin(reticle - pi)
End Sub
White_Click()事件代码如下:
Private Sub White_Click()                    '设置背景色
    Frm.BackColor = QBColor(15)
End Sub
Blue_Click()事件代码如下:
Private Sub Blue_Click()

```

```
Frm.BackColor = QBColor(9)
End Sub
```

Gray_Click()事件代码如下:

```
Private Sub Gray_Click() '设置前景色
    Dim i As Integer
    For i = 1 To 12
        Linret(i).BorderColor = QBColor(8)
    Next i
    LinHH.BorderColor = QBColor(8)
    LinSS.BorderColor = QBColor(8)
    LinMM.BorderColor = QBColor(8)
    Shp.BorderColor = QBColor(8)
End Sub
```

Green_Click()事件代码如下:

```
Private Sub Green_Click()
    Dim i As Integer
    For i = 1 To 12
        Linret(i).BorderColor = QBColor(2)
    Next i
    LinHH.BorderColor = QBColor(2)
    LinSS.BorderColor = QBColor(2)
    LinMM.BorderColor = QBColor(2)
    Shp.BorderColor = QBColor(2)
End Sub
```

Revert_Click()事件代码如下:

```
Private Sub Revert_Click() '恢复背景初始状态颜色
    Dim i As Integer
    Frm.BackColor = &HFFC0C0
    For i = 1 To 12
        Linret(i).BorderColor = QBColor(1)
    Next i
    Shp.BorderColor = QBColor(1)
    LinHH.BorderColor = QBColor(1)
    LinSS.BorderColor = QBColor(1)
    LinMM.BorderColor = QBColor(1)
End Sub
```

Close_Click()事件代码如下:

```
Private Sub Close_Click()
    Unload Me
End Sub
```

(3) 保存窗体, 运行程序, 结果如图 1-1-1 所示。

实验 2 参赛队对阵表生成器

实验题目: 设计一个“参赛队对阵表生成器”程序。

该程序有如下控制功能:

输入参赛队的数目和参赛队的名称, 生成参赛队的对阵表。

程序的运行结果如图 1-2-1 所示。

操作步骤如下。

(1) “指针式时钟”窗体及主要控件属性参照表 1-2-1 设计。




图 1-2-1 参赛队对阵表生成器

表 1-2-1 “参赛队对阵表生成器”窗体及主要控件的属性

对 象	对 象 名	属 性 名	属 性 值	事 件 名
窗体	Frm	Caption	参赛队对阵表生成器	无
		Height	6900	
		Width	7350	
标签	LblInput	Caption	请输入参赛队数:	无
		Left	120	
		Top	240	
	LblName	Caption	请输入参赛队名称:	
		Left	120	
		Top	840	
命令按钮	CmdOk	Caption	OK	Click
		Height	375	
		Width	1215	
		Left	5880	
		Top	240	
	CmdInput	Caption	输入队名	Click
		Height	375	
		Width	1215	
		Left	4440	
		Top	840	
	CmdOutput	Caption	计算对阵表	Click
		Height	375	
		Width	1215	
		Left	5880	
		Top	840	
文本框	TxtNo	Text	4	无

(2) 打开“代码设计”窗口，输入程序代码。

定义窗体变量代码如下：

```
Dim pos As Integer
Dim a() As String
```

CmdInput_Click()事件代码如下：

```
Private Sub CmdInput_Click()
    If pos = UBound(a) Then
```

```

        MsgBox "参赛队已经满了", vbOKOnly + vbCritical, "错误"
    Exit Sub
End If
a(pos) = CboName.Text
CboName.Text = ""
CboName.SetFocus
pos = pos + 1
End Sub

```

CmdOk_Click()事件代码如下:

```

Private Sub CmdOk_Click()
    If TxtNo.Text = "" Then
        MsgBox "请输入参赛队数", vbOKOnly + vbCritical, "错误"
    Exit Sub
    End If
    ReDim a(Val(TxtNo.Text))
    CboName.SetFocus
    CmdInput.Enabled = True
    CmdOutput.Enabled = True
End Sub

```

CmdOutput_Click()事件代码如下:

```

Private Sub CmdOutput_Click()
    Dim i As Integer, j As Integer
    Dim ItemX As ListItem
    Call LstView.ColumnHeaders.Add(1, "_Title", "\")
    For i = 0 To UBound(a) - 1
        Call LstView.ColumnHeaders.Add(i + 2, "_TeamX" & CStr(i), a(i), 2000)
        'Set ItemX = LstView.ListItems.add(i + 1, "_TeamY" & CStr(i), a(i))
    Next i
    For i = 0 To UBound(a) - 1
        Set ItemX = LstView.ListItems.Add(i + 1, "_TeamY" & CStr(i), a(i))
        For j = 0 To i
            If j = i Then
                ItemX.SubItems(j + 1) = "无"
            Else
                ItemX.SubItems(j + 1) = a(i) & " vs " & a(j)
            End If
        Next j
    Next i
End Sub

```

CboName_KeyDown()事件代码如下:

```

Private Sub CboName_KeyDown(KeyCode As Integer, Shift As Integer)
    If KeyCode = 13 Then
        a(pos) = CboName.Text
        CboName.Text = ""
        pos = pos + 1
    End If
End Sub

```

(3) 保存窗体, 运行程序, 结果如图 1-2-1 所示。

实验 3 打印字符图形

实验题目: 设计一个“打印字符图形”程序。

该程序有如下控制功能:

- (1) 可选择打印的形状，菱形或平行四边形；
 (2) 可选择打印的字符。

程序的运行结果如图 1-3-1 所示。




图 1-3-1 打印字符图形

操作步骤如下。

- (1) “指针式时钟”窗体及主要控件属性参照表 1-3-1 设计。

表 1-3-1 “打印字符图形”窗体及主要控件的属性

对 象	对 象 名	属 性 名	属 性 值	事 件 名
窗体	Frm	Caption	打印字符图形	无
		Height	4605	
		Width	3885	
图片框	Pic	AutoRedraw	True	无
		Height	3015	
		Width	3015	
命令按钮	CmdPrint	Caption	打印	Click
		Height	2160	
		Width	2160	
	CmdQuit	Caption	退出	Click
		Height	375	
		Width	975	
		Interval	975	

- (2) 打开“代码设计”窗口，输入程序代码。

CmdPrint_Click()事件代码如下：

```
Private Sub CmdPrint_Click()
    Dim k As Integer, str As String, i As Integer, j As Integer
    Pic.Cls
    k = MsgBox("是否打印菱形，若否，则打印平行四边形", 64 + 3, "提示")
    If k = 2 Then
        Exit Sub
    Else
        str = InputBox("请输入想要打印的字符形状", "输入")
        If k = 6 Then
            For i = 1 To 8
                Pic.Print Tab(17 - i);
```

```

        For j = 1 To i * 2
            Pic.Print str;
        Next j
        Pic.Print
    Next i
    For i = 1 To 8
        Pic.Print Tab(i + 8);
        For j = 1 To 18 - i * 2
            Pic.Print str;
        Next j
        Pic.Print
    Next i
End If
If k = 7 Then
    Pic.Print
    Pic.Print
    Pic.Print
    For i = 1 To 10
        Pic.Print Tab(13 - i);
        For j = 1 To 20
            Pic.Print str;
        Next j
        Pic.Print
    Next i
End If
End If
End Sub

```

CmdQuit_Click()事件代码如下:

```

Private Sub CmdQuit_Click()
    End
End Sub

```

(3) 保存窗体, 运行程序, 结果如图 1-3-1 所示。

实验 4 交换变量的值

实验题目: 设计一个“交换变量的值”程序。

该程序有如下控制功能:

交接 m , n 的值, 不使用临时变量。

程序的运行结果如图 1-4-1 所示。




图 1-4-1 交换变量的值

操作步骤如下。

(1) “指针式时钟”窗体及主要控件属性参照表 1-4-1 设计。

表 1-4-1 “交换 m, n 的值（不使用临时变量）”窗体及主要控件的属性

对 象	对 象 名	属 性 名	属 性 值	事 件 名
窗体	Frm	Caption	交换 m, n 的值（不使用临时变量）	无
		Height	3300	
		Width	4815	
命令按钮	CmdSwap	Caption	交换	Timer
		Height	615	
		Width	1335	

(2) 打开“代码设计”窗口，输入程序代码。

CmdSwap_Click()事件代码如下：

```
Private Sub CmdSwap_Click()
    Dim m As Integer, n As Integer, t As Integer
    m = 32
    n = 8
    Print "交换前: "
    Print "m = " & m & " " & "n = " & n
    Call Swap(m, n)
    Print "交换后: "
    Print "m = " & m & " " & "n = " & n
End Sub
```

Swap()过程代码如下：

```
Private Sub Swap(a As Integer, b As Integer)
    a = a Xor b
    b = a Xor b
    a = a Xor b
End Sub
```

(3) 保存窗体，运行程序，结果如图 1-4-1 所示。

实验 5 复制文件

实验题目：设计一个“文件复制”程序。

该程序有如下控制功能：

输入源文件路径及文件名与目标文件路径名，进行文件的复制。

程序的运行结果如图 1-5-1、图 1-5-2、图 1-5-3 所示。




图 1-5-1 文件复制



图 1-5-2 选择源文件



图 1-5-3 输入目标文件名

操作步骤如下。

(1) “文件复制”窗体及主要控件属性参照表 1-5-1 设计。

表 1-5-1 “文件复制”窗体及主要控件的属性

对 象	对 象 名	属 性 名	属 性 值	事 件 名
窗体	FrmMain	Caption	文件复制	Load
		Height	4935	
		Width	6885	
框架	FraOrigin	Caption	源文件	无
		Height	855	
		Width	5175	
	FraTarget	Caption	目标文件	
		Height	855	
		Width	5175	
命令按钮	CmdOrigin	Caption	浏览...	Click
		Height	255	
		Width	855	
	LinHH	Caption	浏览...	
		Height	255	
		Width	855	
	LinSS	Caption	复制	
		Height	375	
		Width	1215	
	Linret	Caption	退 出	
		Height	375	
		Width	1215	
文本框	TxtOrigin	Text		无
	TxtTarget	Text		
标签	LblNote	Caption	文件复制	无

(2) 打开“代码设计”窗口，输入“文件复制”窗体程序代码。

定义窗体变量代码如下：

```
Option Explicit
Const TextA = "请在此输入源文件路径及文件名"
```

Const TextB = "请在此输入目标文件路径及文件名"

Public FileNameA As String, FileNameB As String '源文件名、目标文件名

Public FilePath As String '源文件路径

CmdOrigin_Click()事件代码如下:

Private Sub CmdOrigin_Click()

FrmOrigin.Show

'打开“选择源文件”窗体

End Sub

CmdTarget_Click()事件代码如下:

Private Sub CmdTarget_Click()

FrmTarget.Show

'打开“输入目标文件名”窗体

End Sub

CmdCopy_Click()事件代码如下:

Private Sub CmdCopy_Click()

'如未输入源文件名,则提示错误

If FileNameA = "" Then

MsgBox "请输入源文件名!", 48, "错误"

Exit Sub

End If

'如未输入目标文件名,则提示错误

If FileNameB = "" Then

MsgBox "请输入目标文件名!", 48, "错误"

Exit Sub

End If

'如源文件不存在,则提示错误

If Dir(FileNameA) = "" Then

MsgBox "源文件不存在!", 48, "错误"

Exit Sub

End If

'如目标文件路径错误,则设置目标文件路径为源文件路径

If Mid(TxtTarget.Text, 2, 1) <> ":" Then

If MsgBox("目标文件路径错误。" & vbCrLf & vbCrLf & "文件将复制到 " & FilePath & ", 目标文件名为 " & TxtTarget.Text & ", 确定吗?", 36, "提示") = 7 Then Exit Sub

If Len(FilePath) = 3 Then

FileNameB = FilePath & TxtTarget.Text

TxtTarget.Text = FileNameB

Else

FileNameB = FilePath & "\" & TxtTarget.Text

TxtTarget.Text = FileNameB

End If

End If

'如源文件与目标文件重名,则提示错误

If FileNameA = FileNameB Then

MsgBox "目标文件与源文件重名,请更改目标文件名!", 48, "错误"

Exit Sub

End If

'如目标文件已存在,则询问是否覆盖

If Dir(FileNameB) <> "" Then

If MsgBox("目标文件已存在,要覆盖现有的文件吗?", 36, "提示") = 7

Then Exit Sub

End If

FileCopy FileNameA, FileNameB

'复制文件

MsgBox "文件复制成功!", 64, "提示"

'提示复制成功

FileNameA = ""

```
FileNameB = ""
TxtOrigin.Text = TextA
TxtTarget.Text = TextB
CmdQuit.SetFocus
End Sub
```

CmdQuit_Click()事件代码如下:

```
Private Sub CmdQuit_Click()
    End
End Sub
```

Form_Click()事件代码如下:

```
Private Sub Form_Click()
    CmdCopy.SetFocus '使“复制”按钮获得焦点
End Sub
```

Form_Load()事件代码如下:

```
Private Sub Form_Load()
    FileNameA = ""
    FileNameB = ""
    FilePath = ""
    TxtOrigin.Text = TextA
    TxtTarget.Text = TextB
End Sub
```

TxtOrigin_Click()事件代码如下:

```
Private Sub TxtOrigin_Click()
    If FileNameA = "" Then TxtOrigin.Text = "" '清除文本框内的提示信息
End Sub
```

TxtOrigin_LostFocus()事件代码如下:

```
Private Sub TxtOrigin_LostFocus()
    FileNameA = TxtOrigin.Text '设置源文件名
    If TxtOrigin.Text = "" Then TxtOrigin.Text = TextA
    '如文件名未输入,则显示提示信息
End Sub
```

TxtTarget_Click()事件代码如下:

```
Private Sub TxtTarget_Click()
    If FileNameB = "" Then TxtTarget.Text = "" '清除文本框内的提示信息
End Sub
```

TxtTarget_LostFocus()事件代码如下:

```
Private Sub TxtTarget_LostFocus()
    FileNameB = TxtTarget.Text '设置目标文件名
    If TxtTarget.Text = "" Then TxtTarget.Text = TextB
    '如文件名未输入,则显示提示信息
End Sub
```

(3) “选择源文件”窗体及主要控件属性参照表 1-5-2 设计。

表 1-5-2 “选择源文件”窗体及主要控件的属性

对 象	对 象 名	属 性 名	属 性 值	事 件 名
窗体	FrmOrigin	Caption	选择源文件	Load
		Height	4020	
		Width	4665	
框架	FraDrv	Caption	驱动器	无
		Height	615	
		Width	1815	